
*set path
cd "/users/andkje/dropbox/wheels/replication" 
set scheme s1mono

global C1="social_group2 social_group3"
global C2="head_educ head_litt occ_agg occ_salary occ_cultivator land rural  Power  INCOME5"
global C3="EW14A  EW14B EW14C EW14D RO5"


*********************************************
**Table 1: Schooling
*********************************************
qui {
use "tab1_sample.dta",clear

*Panel A
eststo regA1: reg class9 F_T_BH F_BH T_BH girl_treated girl treated bihar  [weight=Weight],robust cluster(district)
preserve
keep if e(sample) & F_T_BH==1
predict y_hat
replace y_hat=y_hat-_b[F_T_BH]
sum y_hat
estadd scalar Baseline = r(mean)
restore
estadd local Controls "No"


eststo regA2: reg class9 F_T_BH F_BH T_BH girl_treated girl treated bihar $C1 $C2 [weight=Weight],robust cluster(district)
preserve
keep if e(sample) & F_T_BH==1
foreach v in $C1 $C2  {
qui sum `v' if F_T_BH==1 [weight=Weight] 
qui replace  `v'=r(mean)
}
predict y_hat
replace y_hat=y_hat-_b[F_T_BH]
sum y_hat 
estadd scalar Baseline = r(mean)
restore
estadd local Controls "Yes"

eststo regA3: reg class10 F_T_BH F_BH T_BH girl_treated girl treated bihar [weight=Weight],robust cluster(district)
preserve
keep if e(sample) & F_T_BH==1
predict y_hat
replace y_hat=y_hat-_b[F_T_BH]
sum y_hat
estadd scalar Baseline = r(mean)
restore
estadd local Controls "No"

eststo regA4: reg class10 F_T_BH F_BH T_BH girl_treated girl treated bihar $C1 $C2  [weight=Weight],robust cluster(district)
preserve
keep if e(sample) & F_T_BH==1
foreach v in $C1 $C2   {
qui sum `v' if F_T_BH==1 [weight=Weight] 
qui replace  `v'=r(mean)
}
predict y_hat
replace y_hat=y_hat-_b[F_T_BH]
sum y_hat 
estadd scalar Baseline = r(mean)
restore
estadd local Controls "Yes"

*Panel B
foreach v in F_T_BH F_BH T_BH girl_treated girl treated bihar {
gen `v'X=`v'* Stdmean_distance_05
}

eststo regB1: reg class9 F_T_BHX F_BHX T_BHX girl_treatedX girlX treatedX biharX mean_distance_05 F_T_BH F_BH T_BH girl_treated girl treated bihar [weight=Weight] ,robust cluster(district)
estadd local Controls "No"

eststo regB2: reg class9 F_T_BHX F_BHX T_BHX girl_treatedX girlX treatedX biharX mean_distance_05 F_T_BH F_BH T_BH girl_treated girl treated bihar $C1 $C2 [weight=Weight] ,robust cluster(district)
estadd local Controls "Yes"

eststo regB3: reg class10 F_T_BHX F_BHX T_BHX girl_treatedX girlX treatedX biharX mean_distance_05 F_T_BH F_BH T_BH girl_treated girl treated bihar [weight=Weight] ,robust cluster(district)
estadd local Controls "No"

eststo regB4: reg class10 F_T_BHX F_BHX T_BHX girl_treatedX girlX treatedX biharX mean_distance_05 F_T_BH F_BH T_BH girl_treated girl treated bihar $C1 $C2 [weight=Weight] ,robust cluster(district)
estadd local Controls "Yes"

	label var F_T_BHX "TreatmentXFemaleXBiharXDistance"
	label var F_T_BH "TreatmentXFemaleXBihar"
}

noisily esttab regA1 regA2 regA3 regA4, ///
	keep(F_T_BH)  s(N Baseline Controls) label b(3) se(3)  nomtitles starlevels(* 0.10 ** 0.05 *** 0.01) 
noisily esttab regB1 regB2 regB3 regB4, ///
	keep(F_T_BHX F_T_BH) order(keep(F_T_BHX F_T_BH)) s(N Controls) label b(3) se(3)  nomtitles starlevels(* 0.10 ** 0.05 *** 0.01) 
	
	
	 
*************************************************************
**Table 2: Sample balance
*************************************************************
use "main_sample.dta",clear

qui {
matrix table2=J(17,2,.)
global a=0
foreach v in $C1 $C2 $C3  {
global a= $a + 1
reg `v'  DiD Di bihar  [weight=Weight],robust cluster(district)	
 matrix table2[$a, 1]= _b[DiD]
 matrix table2[$a, 2]= _se[DiD]
}
count 
matrix table2[17, 1]= r(N)
matrix colnames table2 = " " " " 
matrix rownames table2 = "ST/SC/OBC" "Muslim" "Years_of_school_HH_head" "Literate_HH_head" "Agriculture_labour" "Salary_earner" "Cultivator"  "Land_owner" "Rural" "Electricity"  "Income_quintile" "Mother_attended_school" "Father_attended_school" "Mother-in-law_attended_school" "Father-in-law_attended_school" "Age" "Observations"
matrix colnames table2 = "Coefficent" "SE"
}
*Note: add stars manually
mat list table2,format(%9.3f)



	
*************************************************************
**Table 3: Placebo regressions
*************************************************************
qui {
*Panel A
use "main_sample.dta",clear
eststo reg3A1: reg index_decision2005  DiD Di bihar  [weight=Weight],robust cluster(district)
estadd local Controls "No"
eststo reg3A2: reg index_decision2005  DiD Di bihar  $C1 $C2 $C3  [weight=Weight],robust cluster(district)
estadd local Controls "Yes"
eststo reg3A3: reg index_alternative2005 DiD Di bihar    [weight=Weight],robust cluster(district)
estadd local Controls "No"
eststo reg3A4: reg index_alternative2005 DiD Di bihar  $C1 $C2 $C3   [weight=Weight],robust cluster(district)
estadd local Controls "Yes"

*Panel B
use "tab3B_sample.dta",clear
eststo reg3B1: reg index_decision DiD Di bihar  [weight=Weight],robust cluster(district)
estadd local Controls "No"
eststo reg3B2: reg index_decision DiD Di bihar $C1 $C2 $C3    [weight=Weight],robust cluster(district)
estadd local Controls "Yes"
eststo reg3B3: reg index_alternative DiD Di bihar  [weight=Weight],robust cluster(district)
estadd local Controls "No"
eststo reg3B4: reg index_alternative DiD Di bihar $C1 $C2 $C3    [weight=Weight],robust cluster(district)
estadd local Controls "Yes"

label var DiD "GirlTreatedCohortXBihar"
label var Di  "GirlTreatedCohort"
label var bihar "Bihar"
}

noisily esttab reg3A1 reg3A2 reg3A3 reg3A4, ///
	keep(DiD Di bihar) order(DiD Di bihar) s(N Controls) label b(3) se(3)  nomtitles starlevels(* 0.10 ** 0.05 *** 0.01) 
noisily esttab reg3B1 reg3B2 reg3B3 reg3B4, ///
	keep(DiD Di bihar) order(DiD Di bihar) s(N Controls) label b(3) se(3)  nomtitles starlevels(* 0.10 ** 0.05 *** 0.01) 
 


*************************************************************
**Table 4: Main results: The program effect on female empowerment
*************************************************************
qui {
use "main_sample.dta",clear
eststo reg4A1: reg index_decision  DiD Di bihar  [weight=Weight],robust cluster(district)
estadd local Controls "No"
eststo reg4A2: reg index_decision  DiD Di bihar  $C1 $C2 $C3  [weight=Weight],robust cluster(district)
estadd local Controls "Yes"
eststo reg4A3: reg index_alternative  DiD Di bihar  [weight=Weight],robust cluster(district)
estadd local Controls "No"
eststo reg4A4: reg index_alternative DiD Di bihar  $C1 $C2 $C3   [weight=Weight],robust cluster(district)
estadd local Controls "Yes"
label var DiD "GirlTreatedCohortXBihar"
label var Di  "GirlTreatedCohort"
label var bihar "Bihar"
}
noisily esttab reg4A1 reg4A2 reg4A3 reg4A4, ///
	keep(DiD Di bihar) order(DiD Di bihar) s(N Controls) label b(3) se(3)  nomtitles starlevels(* 0.10 ** 0.05 *** 0.01) 

	

*************************************************************
**Table 5: Individual empowerment outcomes
*************************************************************	
qui {
use "main_sample.dta",clear
matrix table5=J(26,4,.)
gen n=_n
gen p=0
global a=0


foreach var in ew_Decides1 ew_Decides2 ew_Decides3 ew_Decides4 ew_Decides5 ew_Decides6 ew_Decides7 ew_Decides8  ew_mobility9 ew_mobility10 ew_mobility11 ew_mobility12  ew_memberA ew_memberB ew_memberC ew_memberD ew_meeting ew_discussA ew_discussB ew_discussC  ew_decideswork ew_currentwork  ew_eat ew_cash ew_loan ew_bank    {
	
global a=$a +1

qui reg `var' DiD Di bihar   $C1 $C2 $C3  [weight=Weight]
qui sum `var'  if e(sample)==1 & Di==0  [weight=Weight]
replace `var' =(`var'-r(mean))/r(sd)  

reg `var' DiD Di bihar   $C1 $C2 $C3  [weight=Weight]
qui {
matrix table5[$a ,1 ]= _b[DiD]
matrix table5[$a  ,2 ]= _se[DiD]
matrix table5[$a  ,4 ]= e(N)
qui test DiD
replace p= r(p) if  n==$a
eststo reg$a
}
}
matrix rownames table5 = "What_to_cook_on_a_daily_basis" "Whether_to_buy_an_expensive_item"  "Number_of_children_to_have"  "What_to_do_if_sick"  "Whether_to_buy_land_or_property" "Expenses_on_social_functions"  "What_to_do_if_a_child_falls_sick"    "To_whom_your_child_should_marry"  "Local_health_centre" "Home_of_relatives_or_friends"  "Local_grocery_shop"  "Short_distance_by_train_or_bus"   "Member_of_womens_group"  "Member_of_self_help_group"  "Member_of_savings_group"   "Member_of_political_organization"  "Attend_public_meeting_last_year"   "Things_from_work_or_farm"  "What_to_spend_money_on"  "Village_issues_or_politics"  "Most_say_about_own_work"  	"Currently_working"	   "Eats_together_with_husband"  "Cash_at_hand"  "Name_on_ownership_of_home"  "Name_on_bank_account"
	 

*Make FDR q-values
drop n
gen n=_n
global a=0
foreach v in ew_Decides1 ew_Decides2 ew_Decides3 ew_Decides4 ew_Decides5 ew_Decides6 ew_Decides7 ew_Decides8  ew_mobility9 ew_mobility10 ew_mobility11 ew_mobility12  ew_memberA ew_memberB ew_memberC ew_memberD ew_meeting ew_discussA ew_discussB ew_discussC  ew_decideswork ew_currentwork  ew_eat ew_cash ew_loan ew_bank{
	global a=$a +1
	reg `v' DiD Di bihar   $C1 $C2 $C3  [weight=Weight]
	qui test DiD
	replace p= r(p) if  n==$a
}
keep p n

global no=26
keep if n<=$no

matrix p=J($no ,1,.)
forvalues n=1/$no {
	sum p if n==`n'
	matrix p[`n',1]= r(mean)
}
do "ming.do"
minq p, q(q)
svmat q

global a=0
forvalues a=1/$no {
	global a=$a +1
	qui sum q if n==`a'
	matrix table5[$a ,3]= r(mean)
	
}
matrix colnames table5 = "Coefficent" "SE" "FDR q-value" "N"
}

*Note: add stars manually
mat list table5,format(%9.3f)



	

*************************************************************
**Table 6: Robustness specifications
*************************************************************
qui {
use "tab6_sample.dta",clear

*Only rural
eststo reg6A1: reg index_decision DiD Di bihar   $C1 $C2 $C3 if  rural==1 & east_group==1  [weight=Weight],robust cluster(district)
eststo reg6B1: reg index_alternative DiD Di bihar   $C1 $C2 $C3 if rural==1 & east_group==1 [weight=Weight],robust cluster(district)

*UP+MP as control group
gen MP_UP=STATEID==10|STATEID==23|STATEID==9
eststo reg6A2: reg index_decision DiD Di bihar   $C1 $C2 $C3 if   MP_UP==1 [weight=Weight],robust cluster(district)	
eststo reg6B2: reg index_alternative DiD Di bihar   $C1 $C2 $C3 if  MP_UP==1 [weight=Weight],robust cluster(district)	

*Rest of India as control group
eststo reg6A3: reg index_decision  DiD Di bihar   $C1 $C2 $C3  [weight=Weight],robust cluster(district)	
eststo reg6B3: reg index_alternative DiD Di bihar   $C1 $C2 $C3  [weight=Weight],robust cluster(district)	

*No Control group
eststo reg6A4: reg index_decision DiD   $C1 $C2 $C3 if    bihar==1   [weight=Weight],robust cluster(district)	
eststo reg6B4: reg index_alternative DiD Di bihar   $C1 $C2 $C3 if  bihar==1  [weight=Weight],robust cluster(district)	

*Narrow cohorts
foreach v in 1 2 3  {
	global a=4 + `v'
replace Di=sum_girl_treated_alt`v'
replace DiD=Di*bihar
replace Di=. if sum_girl_treated_alt`v'!=1 & sum_girl_control_alt`v'!=1
eststo reg6A$a: reg index_decision  DiD Di bihar   $C1 $C2 $C3 if  east_group==1  [weight=Weight],robust cluster(district)	
eststo reg6B$a: reg index_alternative DiD Di bihar   $C1 $C2 $C3 if  east_group==1 [weight=Weight],robust cluster(district)	
}

label var DiD "GirlTreatedCohortXBihar"
}

forvalues i=1/7 {
noisily esttab reg6A`i' reg6B`i', ///
	keep(DiD) order(DiD) s(N) label b(3) se(3) wide  nomtitles starlevels(* 0.10 ** 0.05 *** 0.01) 
}


	
	
************************************************
**Table 7: Interaction with mothers
************************************************
qui {
use "main_sample.dta",clear

qui gen DiD2=DiD * Mother 
qui gen Di2=Di * Mother 
qui gen bihar2=bihar * Mother 

eststo reg7A1: reg index_decision  DiD2 DiD Di2 Di bihar2 bihar Mother $C1 $C2 $C3  [weight=Weight],robust cluster(district)
eststo reg7A2: reg index_alternative  DiD2 DiD Di2 Di bihar2 bihar Mother  $C1 $C2 $C3  [weight=Weight] ,robust cluster(district)
label var DiD2  "GirlTreatedCohortXBiharXMothers"
label var DiD  "GirlTreatedCohortXBihar"
}
noisily esttab reg7A1 reg7A2 , ///
	keep(DiD2 DiD) order(DiD2 DiD) s(N) label b(3) se(3)  nomtitles starlevels(* 0.10 ** 0.05 *** 0.01) 
	



*************************************************************
**Table 8: Alternative specifications
*************************************************************	

qui {
*Panel A: Women FEs
use "tab8A_sample.dta",clear 
xtset individual_id
eststo reg8A1: xtreg  index_decision_panel  tripple sum_girl_treatedPost biharPost post  [weight=Weight] ,robust cluster(district) fe
eststo reg8A2: xtreg  index_alternative_panel  tripple sum_girl_treatedPost  biharPost post [weight=Weight] ,robust cluster(district) fe

*Panel B: girls versus boys in treatment and control cohorts
use "tab8B_sample.dta",clear 
eststo reg8B1: reg index_decision tripple sum_girl_treated sum_treated sum_treated_bihar sum_girl sum_girl_bihar  bihar $C1 $C2  $C3   [weight=Weight],robust cluster(district)
eststo reg8B2: reg index_alternative tripple sum_girl_treated sum_treated sum_treated_bihar sum_girl sum_girl_bihar  bihar $C1 $C2  $C3   [weight=Weight],robust cluster(district)
}
qui label var tripple "GirlTreatedCohortXBiharXPost"

**Panel A: Panel regression, mother fixed effects
noisily esttab reg8A1 reg8A2 , ///
	keep(tripple) order(tripple) s(N) label b(3) se(3)  nomtitles starlevels(* 0.10 ** 0.05 *** 0.01) 

qui label var tripple "GirlXTreatedCohortXBihar"
	
**"Panel B: Triple-difference regression, women with boys"	
noisily esttab reg8B1 reg8B2 , ///
	keep(tripple) order(tripple) s(N) label b(3) se(3)  nomtitles starlevels(* 0.10 ** 0.05 *** 0.01) 	


				
*************************************************************
**Table 9: Mechanisms
*************************************************************	 	 
qui {
use "main_sample.dta",clear
gen DiD2=0
gen Di2=0
gen bihar2=0
gen std=0

foreach v in Stdvillage_pref_boys_05 village_decide_05 Stdmean_distance_05 index_amenity_distance_05 {
qui replace  DiD2=DiD * `v'
qui replace  Di2=Di * `v'
qui replace  bihar2=bihar * `v'
replace std=`v'
eststo `v'1: reg index_decision  DiD2 DiD Di2 Di bihar2 bihar std  $C1 $C2 $C3  [weight=Weight],robust cluster(district)
eststo `v'2: reg index_alternative  DiD2 DiD Di2 Di bihar2 bihar std  $C1 $C2 $C3  [weight=Weight],robust cluster(district)
label var DiD  "GirlTreatedCohort $\times$ Bihar"
}
 
} 
*Panel A: Preferences for boys (village-level 2005)
label var DiD2  "GirlTreatedCohort $\times$ Bihar $\times$ Boy preferences (SD)"
noisily esttab Stdvillage_pref_boys_051 Stdvillage_pref_boys_052 , ///
	keep(DiD2  DiD) order(DiD2  DiD) s(N) label b(3) se(3)  nomtitles starlevels(* 0.10 ** 0.05 *** 0.01) 
 
*Panel B: Intra-household decision-making (village-level 2005)
label var DiD2  "GirlTreatedCohort $\times$ Bihar $\times$ Decision-making (SD)"
noisily esttab village_decide_051 village_decide_052 , ///
	keep(DiD2  DiD) order(DiD2  DiD) s(N) label b(3) se(3)  nomtitles starlevels(* 0.10 ** 0.05 *** 0.01) 
 
*Panel C: Distance to secondary school (village-level 2005)
label var DiD2  "GirlTreatedCohort $\times$ Bihar $\times$ School distance (SD)"
noisily esttab Stdmean_distance_051 Stdmean_distance_052, ///
	keep(DiD2  DiD) order(DiD2  DiD) s(N) label b(3) se(3)  nomtitles starlevels(* 0.10 ** 0.05 *** 0.01) 

*Panel D: Distance to secondary school (village-level 2005)
label var DiD2  "GirlTreatedCohort $\times$ Bihar $\times$ Amenities (SD)"	
noisily esttab index_amenity_distance_051 index_amenity_distance_052 , ///
	keep(DiD2  DiD) order(DiD2  DiD) s(N) label b(3) se(3)  nomtitles starlevels(* 0.10 ** 0.05 *** 0.01) 
 


*****************************************************************
**Table A3: Program effects on addition outcomes
*****************************************************************
qui {
use "main_sample.dta",clear
duplicates drop hh_id,force
matrix tableA1=J(14,4,.)
matrix p=J(40,1,.)
global a=0
gen p=0
gen n=_n
foreach var in totexp foodexp med_health_exp radioW newspaperW radioM newspaperM index_conf dowry {	
	global a= $a + 1

	qui sum `var'  if Di==0  [weight=Weight] 
	qui replace `var' =(`var' -r(mean))/r(sd) 
	
	reg `var' DiD Di bihar $C1 $C2 [weight=Weight],robust cluster(district)
	qui {
	matrix tableA1[$a, 1]= _b[DiD]
	matrix tableA1[$a , 2]= _se[DiD]
	matrix tableA1[$a , 4]= e(N)
	
	qui test DiD
	replace p= r(p) if  n==$a
	matrix p[$a , 1]= r(p)
	}
}

****Women-level
use "main_sample.dta",clear
gen p=0
gen n=_n
foreach var in health_beliefs1  health_beliefs2  aids pref_fertility pref_boys{
	global a= $a + 1
	qui sum `var'  if Di==0  [weight=Weight] 
	qui replace `var' =(`var' -r(mean))/r(sd) 
	reg `var' DiD Di bihar $C1 $C2 $C3  [weight=Weight],robust cluster(district)
	qui {
	matrix tableA1[$a, 1]= _b[DiD]
	matrix tableA1[$a , 2]= _se[DiD]
	matrix tableA1[$a , 4]= e(N)
	
	qui test DiD
	replace p= r(p) if  n==$a
	matrix p[$a , 1]= r(p)
	}
}

**FDR q-values       
keep n  
svmat p
keep if n<=$a
 do  "ming.do"
minq p, q(q)
svmat q, names(q)

global n=$a *2
global b= $a
global a=0
forvalues i=1/$b {
	global a=$a + 1
	sum q1 if n==`i'
	matrix tableA1[$a , 3]= r(mean)
}

	matrix rownames tableA1  = "Consumption expenditure "  "Food"  "Medical and education"  "Radio(Women)"  "Newspaper(Women)"  "Radio(Men)"  "Newspaper(Men)"   "Confidence institutions (index)"   "Typical cash amount as dowry"  "General health beliefs" "Reproductive health beliefs"  "HIV/AIDS awareness"  "Ideal number of kids"   "Preference for boys" 
matrix colnames table5 = "Coefficent" "SE" "FDR q-value" "N"
}

*Note: ADD STARS MANUALLY
mat list tableA1,format(%9.3f)





*************************************************************
**Figure 1: Graphical assessment of parallel trend assumption
*************************************************************	 
qui {
use "fig1_sample.dta",clear

gen n=1
gen year_no=sum(n)
gen nobihar=bihar==0

foreach i in 1 2 {
gen coeff_bihar`i'=.
gen coeff_other`i'=.
gen upper_b`i'=.
gen lower_b`i'=.
gen upper_o`i'=.
gen lower_o`i'=.
}

global a=0
foreach c in 1112 1314 1516 1718 1920 {
	qui {
	global a=$a +2
	
	reg index_decision nobihar bihar if  sum_girl_cohort_`c'==1 & age2006>(15 + $a )  [weight=Weight],noconst
	replace coeff_bihar1=_b[bihar] if year_no==(1996 - $a)
	replace upper_b1=_b[bihar] + (1.96*_se[bihar]) if year_no==(1996 - $a)
	replace lower_b1=_b[bihar]- (1.96*_se[bihar]) if year_no==(1996 - $a)
	replace coeff_other1=_b[nobihar] if year_no==(1996 - $a)
	replace upper_o1=_b[nobihar] + (1.96*_se[nobihar]) if year_no==(1996 - $a)
	replace lower_o1=_b[nobihar]- (1.96*_se[nobihar]) if year_no==(1996 - $a)
	
	reg index_alternative nobihar bihar if  sum_girl_cohort_`c'==1 & age2006>(15 + $a )  [weight=Weight],noconst
	replace coeff_bihar2=_b[bihar] if year_no==(1996 - $a)
	replace upper_b2=_b[bihar] + (1.96*_se[bihar]) if year_no==(1996 - $a)
	replace lower_b2=_b[bihar]- (1.96*_se[bihar]) if year_no==(1996 - $a)
	replace coeff_other2=_b[nobihar] if year_no==(1996 - $a)
	replace upper_o2=_b[nobihar] + (1.96*_se[nobihar]) if year_no==(1996 - $a)
	replace lower_o2=_b[nobihar]- (1.96*_se[nobihar]) if year_no==(1996 - $a)
	}
}
duplicates drop year_no bihar,force
sort year_no
}
twoway con coeff_bihar1 year_no if year_no>=1986 & year_no<=1994,color(black) || rcap  upper_b1  lower_b1  year_no if year_no>=1986 & year_no<=1994,color(black) || con coeff_other1 year_no if year_no>=1986 & year_no<=1994, color(gray) || rcap  upper_o1  lower_o1  year_no if year_no>=1986 & year_no<=1994,color(gray) subtitle("Intra-household decision making")  xtitle("") yscale(range(-0.8 .6)) ylabel(-.8(.2).6)     xscale(range(1985 1995)) xlabel(1984(2)1994)    legend(order(1 "Bihar" 3 "Control states"))  xline(1991) xlabel(1986 "1986-87" 1988 "1988-89" 1990 "1990-91" 1992 "1992-93" 1994 "1994-95")


twoway con coeff_bihar2 year_no if year_no>=1986 & year_no<=1994,color(black) || rcap  upper_b2  lower_b2  year_no if year_no>=1986 & year_no<=1994,color(black) || con coeff_other2 year_no if year_no>=1986 & year_no<=1994, color(gray) || rcap  upper_o2  lower_o2  year_no if year_no>=1986 & year_no<=1994,color(gray) subtitle("Alternative measure")  xtitle("") yscale(range(-0.8 .6)) ylabel(-.8(.2).6)     xscale(range(1985 1995)) xlabel(1984(2)1994)    legend(order(1 "Bihar" 3 "Control states"))  xline(1991) xlabel(1986 "1986-87" 1988 "1988-89" 1990 "1990-91" 1992 "1992-93" 1994 "1994-95")




